5. Korlátozott kibocsátású Cournot-duopólium

Ebben a részben olyan Cournot féle duopol modellt vizsgálunk, ahol az egyes vállalatok által kibocsátott termékek mennyisége korlátozva van, Tönu Puu, Anna Norin: *"Cournot duopoly when the competitors operate under capacity constraints"* ([7], *Chaos, Solitons and Fractals 18 (2003) 577-592*) dolgozat eredményei alapján. Keressük a Cournot-megoldásokat, vizsgáljuk ezek stabilitását, valamint példán keresztül szemléltetjük a modell viselkedését bizonyos paraméterértékek esetén.

5.1. Korlátozott kibocsátású modell

Ebben a részben hiperbolikus árfüggvénnyel dolgozunk, $d = 1$, $\epsilon = 1$ paraméterek mellett:

$$P(Q) = \frac{1}{Q},$$

ahol $Q = q_1 + q_2$. Vezessük be a kibocsátott mennyiségek maximális értékeit, ezeket jelölje u az első és v a második cég esetén. Továbbá $Q \leq u + v$.

Tegyük fel továbbá, hogy a cégek teljes költségfüggvényei a következő alakban állnak elő:

$$\mathrm{TC}_1(q_1) = -\ln\left(1 - \frac{q_1}{u}\right), \quad \mathrm{TC}_2(q_2) = -\ln\left(1 - \frac{q_2}{v}\right).$$



Az egyes cégek teljes költségfüggvényei

1. ábra

Ekkor ha a kibocsátás zéró, a költségfüggvények is nullák lesznek, és ha a kibocsátás eléri korlátját, az első cég esetében u-t a másodiknál pedig v-t, akkor a vállalatok teljes költségfüggvényei végtelenbe tartanak. Ekkor a teljes bevételek:

$$\mathrm{TR}_1 = P(Q)\, q_1 = \frac{q_1}{q_1+q_2}, \quad \mathrm{TR}_2 = P(Q)\, q_2 = \frac{q_2}{q_1+q_2}$$

A cégek profitjai legyenek:

$$\pi_1 = \mathrm{TR}_1 - \mathrm{TC}_1 = \frac{q_1}{q_1+q_2} + \ln\left(1 - \frac{q_1}{u}\right), \quad \pi_2 = \mathrm{TR}_2 - \mathrm{TC}_2 = \frac{q_2}{q_1+q_2} + \ln\left(1 - \frac{q_2}{v}\right).$$

```
Clear[Price, u, v, TC1, TC2, TR1, TR2, π1, π2, sol1, sol2]
```

```
Price = 1 / (q1 + q2);
TC1 = -Log[1 - q1 / u]; TC2 = -Log[1 - q2 / v];
TR1 = Price * q1; TR2 = Price * q2;
π1 = TR1 - TC1 ;
π2 = TR2 - TC2;
```

A vállalatok reakciógörbéi (lásd a 2. fejezetet):

```
sol1 = Simplify[Solve[∂q1 π1 == 0, q1],
  Assumptions → {q1 ≥ 0, q2 ≥ 0, u ≥ 0, v ≥ 0}]
sol2 = Simplify[Solve[∂q2 π2 == 0, q2],
  Assumptions → {q1 ≥ 0, q2 ≥ 0, u ≥ 0, v ≥ 0}]
```

$$\left\{\left\{q1 \to -\frac{3\, q2}{2} - \frac{1}{2}\sqrt{q2\,(5\, q2 + 4\, u)}\right\},\right.$$
$$\left.\left\{q1 \to \frac{1}{2}\left(-3\, q2 + \sqrt{q2\,(5\, q2 + 4\, u)}\right)\right\}\right\}$$

Az interaktív ábra mutatja az egyes cégek isoprofit görbéit, vastag vonalakkal kiemelve a Cournot-megoldáshoz tartozó görbéket. Az illusztrációban a maximálisan termelhető mennyiségek szabadon változtathatóak.



2. ábra

Mindkét vállalat esetén két megoldást kapunk, melyekből az első kizárható, hiszen nemnegatív kibocsátásokkal dolgozunk. Ezért:

$$\begin{aligned} q_1 &= f(q_2) = \frac{1}{2}\left(-3\,q_2 + \sqrt{q_2\,(5\,q_2 + 4\,u)}\right), \\ q_2 &= g(q_1) = \frac{1}{2}\left(-3\,q_1 + \sqrt{q_1\,(5\,q_1 + 4\,v)}\right). \end{aligned} \tag{1}$$

Ez az egyenlet tulajdonképpen felfogható úgy, mint egy diszkrét dinamikus rendszer, ahol a $(t+1)$-edik időpontbeli kibocsátott mennyiségek a t-edik időpontbeliek alapján határozhatók meg.

$$\begin{aligned} q_1(t+1) &= \frac{1}{2}\left(-3\,q_2(t) + \sqrt{q_2(t)\,(5\,q_2(t) + 4\,u)}\right), \\ q_2(t+1) &= \frac{1}{2}\left(-3\,q_1(t) + \sqrt{q_1(t)\,(5\,q_1(t) + 4\,v)}\right). \end{aligned} \tag{2}$$

A Cournot-megoldás, ahol a két reakciógörbe metszi egymást, azaz:

```
egyensuly =
 FullSimplify[
  Solve[{q1 == 1/2 (-3 q2 + Sqrt[q2 (5 q2 + 4 u)]),
    q2 == 1/2 (-3 q1 + Sqrt[q1 (5 q1 + 4 v)])}, {q1, q2}],
  Assumptions → {u ≥ 0, v ≥ 0}]
```

$$\left\{\{q1\to 0,\ q2\to 0\},\ \left\{q1\to \frac{u^2\, v}{u^2+3\, u\, v+v^2},\ q2\to \frac{u\, v^2}{u^2+3\, u\, v+v^2}\right\}\right\}$$

Az egyenletrendszer fixpontjai, azaz egy-periodikus pontjai definíció szerint pontosan ezek a pontok lesznek.

Megjegyezzük, hogy ha $q_2 > u$, vagy $q_1 > v$, akkor (2) alapján a vállalatok reakciófüggvényei negatív értékeket vesznek fel, aminek gyakorlatban nincs értelme, tehát a reakciógörbék a (2) egyenletekben leírtaktól annyiban módosulnak, hogy $q_2 > u$, illetve $q_1 > v$ esetén 0 értéket vesznek fel.

5.1.1. A Cournot-megoldás stabilitása

A (2) rendszer esetén a Cournot-megoldás stabilitása a diszkrét dinamikai rendszerek elmélete alapján a jobb oldal Jacobi mátrixa sajátértékeinek nagyságától függ. Ha ezek a sajátértékek a Cournot-megoldás helyén véve egynél kisebb abszolút értékűek, akkor a fixpont aszimptotikusan stabilis [19].

```
egyensuly
```

$$\left\{\{q1\to 0,\ q2\to 0\},\ \left\{q1\to \frac{u^2\, v}{u^2+3\, u\, v+v^2},\ q2\to \frac{u\, v^2}{u^2+3\, u\, v+v^2}\right\}\right\}$$

```
FullSimplify[{q1, q2} /. Solve[
   FullSimplify[{q1 /. sol1[[2]], q2 /. sol2[[2]]}] ==
    {q1, q2}, {q1, q2}],
 Assumptions → {u ≥ 0, v ≥ 0, q1 ≥ 0, q2 ≥ 0}]
```

$$\left\{\{0,\ 0\},\ \left\{\frac{u^2\, v}{u^2+3\, u\, v+v^2},\ \frac{u\, v^2}{u^2+3\, u\, v+v^2}\right\}\right\}$$

A (2) rendszer Jacobi mátrixa:

```
Clear[JM, JM2, JM2C];
```

```
JM = Simplify[D[FullSimplify[
      {q1 /. sol1[[2]], q2 /. sol2[[2]]}], {{q1, q2}}]];
(JM2 = FullSimplify[JM, Assumptions →
      {u ≥ 0, v ≥ 0, q1 ≥ 0, q2 ≥ 0}]) // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{1}{2}\left(-3+\frac{5\,q2+2\,u}{\sqrt{q2\,(5\,q2+4\,u)}}\right) \\ \frac{1}{2}\left(-3+\frac{5\,q1+2\,v}{\sqrt{q1\,(5\,q1+4\,v)}}\right) & 0 \end{pmatrix}$$

Ha $q_1 = q_2 = 0$, akkor a vállalatok nem bocsátanak piacra termékeket, az árfüggvény pedig a végtelenbe tart ha $q_1, q_2 \to 0$, így ezt az egyensúlyi helyzetet nem vizsgáljuk.

A második eset stabilitásának vizsgálata érdekesebb, amikor is a Cournot-megoldás az alábbi alakot ölti:

$$q_1 = \frac{u^2 v}{u^2+3\,u\,v+v^2},\; q_2 = \frac{u\,v^2}{u^2+3\,u\,v+v^2}\,.$$

A Jacobi mátrix ekkor:

```
(JM2C = FullSimplify[JM2 /. egyensuly[[2]],
    Assumptions → {u ≥ 0, v ≥ 0}]) // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{u^2-v^2}{2\,u\,v+3\,v^2} \\ \frac{-u^2+v^2}{u\,(3\,u+2\,v)} & 0 \end{pmatrix}$$

melynek sajátértékei:

```
Clear[Ev1, Ev2]
```

```
Ev1 = Eigenvalues[JM2C][[1]]
```

$$-\frac{i\,\left(u^2-v^2\right)}{\sqrt{6\,u^3\,v+13\,u^2\,v^2+6\,u\,v^3}}$$

```
Ev2 = Eigenvalues[JM2C][[2]]
```

$$\frac{i\,\left(u^2-v^2\right)}{\sqrt{6\,u^3\,v+13\,u^2\,v^2+6\,u\,v^3}}$$

Mivel a két sajátérték egymásnak mínusz egyszerese, így a stabilitás meghatározásánál elegendő az egyik sajátérték abszolút értékét vizsgálni.

A megoldás stabilitását a következő ábra szemléleti:



3. ábra

Az ábrán a vastag fekete görbék mutatják azokat az értékeket, ahol a sajátértékek egy abszolút értékűek. Az ábráról leolvasható, hogy ha az egyik cég maximális kibocsátása megközelítőleg több mint nyolcszorosa a másik cég maximális kibocsátásának, akkor a Cournot-megoldás instabillá válik.

Vizsgáljuk most a Cournot-megoldás stabilitását numerikusan a *Mathematica* segítségével. Mivel a Jacobi mátrix főátlójában nullák szerepelnek, így a sajátértékek vizsgálata helyett a mátrix determinánsát is vizsgálhatjuk, ugyanis egy

```
Clear[M];
```

```
M = {{0, a}, {b, 0}};
MatrixForm[M]
```

$$\begin{pmatrix} 0 & a \\ b & 0 \end{pmatrix}$$

alakú mátrix sajátértékei

```
Eigenvalues[M]
```

$\{-\sqrt{a}\ \sqrt{b},\ \sqrt{a}\ \sqrt{b}\}$

és determinánsa:

```
Det[M]
```

$-a\,b$

Tehát a stabilitás szempontjából elegendő a (2) jobb oldala Jacobi mátrixának determinánsát vizsgálni.

A determináns értéke a Cournot-megoldás helyén:

```
Clear[JDet];
```

```
JDet = FullSimplify[Assuming[u ≥ 0 && v ≥ 0, Det[JM2C]]]
```

$$\frac{\left(u^2 - v^2\right)^2}{u\, v\, (3\, u + 2\, v)\, (2\, u + 3\, v)}$$

A stabilitás meghatározásához a *JDet* = *1* egyenletet kell megoldani:

```
Clear[solRed];
```

```
solRed = Assuming[u ≥ 0 && v ≥ 0,
  FullSimplify[Reduce[JDet ≤ 1, {u, v}, Reals]]]
```

$$u > 0 \;\&\&\; \frac{1}{2}\left(-1 + \sqrt{2}\right)\left(-3 + \sqrt{13}\right) u \le v \le \frac{1}{2}\left(1 + \sqrt{2}\right)\left(3 + \sqrt{13}\right) u$$

```
solRed // N
```

$$u > 0. \;\&\&\; 0.125414\, u \le v \le 7.97361\, u$$

A (3) ábrával összhangban kaptuk, hogy a Cournot-megoldás akkor válik instabillá, ha az egyik cég maximális kibocsátása majdnem nyolcszorosa a másikénak.

Ezek alapján a következő két példa szemlélteti a megoldások viselkedését a maximális kibocsátások különböző arányai esetén.

1. PÉLDA

Ebben az esetben egyenlő maximális kibocsátás értékeket adunk a modellnek. A stabilitásvizsgálat értelmében ekkor a Cournot-megoldás aszimptotikusan stabilis lesz. Legyenek a maximális kibocsátások az alábbiak:

```
Clear[u, v, var, f, IC, NN, sol];
u = 10; v = 10;
```

Ekkor a diszkrét rendszer a $(q_1,\ q_2) = (4,\ 2);\ (1,\ 1)$ kezdeti értékek esetén:

```
Clear[DESolve];
DESolve[f_, IC_?MatrixQ, n_Integer] :=
 Map[NestList[f, #, n] &, #] & [IC]
```

```
var = {q1, q2};
f[{t_, q1_, q2_}] := {t + 1, 1/2 (-3 q2 + Sqrt[q2 (5 q2 + 4 u)]),
   1/2 (-3 q1 + Sqrt[q1 (5 q1 + 4 v)])}
```

```
IC = {{0., 4., 2.},
    {0., 1., 1.}, {0., 5., 2.}, {0., 3., 1.}};
NN = 5; sol = DESolve[f, IC, NN];
```



4. ábra

Az ábra alapján az eltérő kezdeti értékekből induló megoldások mindkét esetben az egyensúlyhoz tartanak.

2. PÉLDA

Ebben a példában az első cég maximális kibocsátását a második cég maximális kibocsátásának nyolcszorosára választjuk. A kezdeti értékeket nem módosíatjuk. Nézzük meg ekkor a rendszer viselkedését! A rendszer ekkor instabilitást mutat.

```
Clear[u, v, var, f, IC, NN, sol]
```

```
u = 80; v = 10;
```

```
var = {q1, q2};
f[{t_, q1_, q2_}] := {t + 1, 1/2 (-3 q2 + Sqrt[q2 (5 q2 + 4 u)]),
   1/2 (-3 q1 + Sqrt[q1 (5 q1 + 4 v)])}
```

```
IC = {{0., 4., 2.},
   {0., 1., 1.}, {0., 5., 2.}, {0., 3., 1.}};
NN = 10; sol = DESolve[f, IC, NN];
```



5. ábra

Interaktív kísérletek: Állandó stratégia

Vizsgáljuk meg a modellt eltérő maximális kibocsátott mennyiségek esetén! Az alábbi illusztráció az eddigi fejezetekben megszokott módon kirajzolja a vállalatok költségfüggvényeit, az árfüggvényt és ezek mellett vagy a reakció függvényeket, az isoprofit görbéket, és a Cournot-megoldást vagy pedig az egyes vállalatok profitfüggvényeit.

Felhasználói javaslat:

– A program használata során vigyázzunk, hogy az általunk választott kezdeti mennyiségek, mindig kisebbek legyenek, mint a vállalatok által maximálisan kibocsátható mennyiségek: $q_1 < u$ és $q_2 < v$.

Korlátozott kibocsátású modell: Versengés

Keresleti súly: $Q = w\,q_1 + (1-w)q_2$

w 0.5

Árfüggvény: $P(Q) = d\,Q^{-\epsilon}$

d 1. ϵ 1.

Első vállalat költségfüggvénye: $-\log\left(1-\frac{q_1}{u}\right)$

Második vállalat költségfüggvénye: $-\log\left(1-\frac{q_2}{v}\right)$

u 10. v 10.

Full PlotRange

Kezdeti mennyiségek

Profitfüggvények | Cournot–megoldás

TC=FC+VC



Árfüggvény:P(Q)





Verseny esetén

0.594535 + 0.594535 =

1.18907 profit

0.3 eladási áron

3.33333 és 3.33333 mennyiségekkel

realizálható.

Speciális esetek

1. KÍSÉRLET

Dolgozzunk az alábbi kezdeti beállításokkal:

$w = 0.5$,
$d = 1$, $\epsilon = 1$,
$u = 10$, $v = 10$.

Ha növeljük a valamelyik cég által kibocsátható mennyiséget ($v = 60,\ 80$), akkor nyilvánvalóan az a cég fog magasabb profitot realizálni a Cournot-megoldás értékénél. Az eladási ár ekkor fokozatosan csökken. Ha a cégek eltérő arányban vannak jelen a piacon, akkor az egyensúlyban az a vállalat jár jobban, amelyik a piacnak kisebb részét birtokolja, de többet tud termelni. Az $\epsilon = 0.5$ paraméter választása esetén a profitok ugyan magasabbak, de hasonlóan alakulnak. Az árak valamivel magasabbak, mint az $\epsilon = 1$ esetén. Az árrugalmas kereslet esetén (pl. : $\epsilon = 1.125$) szintén hasonló a helyzet.

Interaktív kísérletek: Diszkrét stratégia

A következő interaktív illusztráció a 3. fejezetben bemutatott diszkrét interaktív illusztrációhoz hasonló módon mutatja a Cournot-megoldást különböző kezdeti értékek mellett. Az illusztráció segítségével következtethetünk az egyensúlyi helyzet stabilitására.


Korlátozott kibocsátású diszkrét modell: Versengés
Keresleti súly: Q=w q1+(1−w)q2
w 0.5
Árfüggvény: P(Q)=d Q−ϵ
d 1
ϵ 1
Első vállalat költségfüggvénye:
−log(1−q1/u)
u 10
Második vállalat költségfüggvénye:
−log(1−q2/v)
v 21.
Iteráció 10
Max. kezdeti értékek:
q1 5
q2 5
PlotRange: 10
Költségfüggvények: TC
TC
q
Árfüggvény:P(Q)
P(Q)
Q
Egyensúlyi helyzet:
(3.04127, 6.38668)
π1= 0.282573 π2= 0.99225

Speciális esetek

2. Kísérlet

Vizsgáljuk a megoldást az alábbi paraméter beállítások mellett:

$w = 0.5$,
$d = 1, \epsilon = 1$,
$u = 10, v = 10$.

A megoldás ekkor asszimtpotikusan stabil. A keresleti súly, illetve a kereslet árrugalmassága nem befolyásolja a megoldás stabilitását. Az 5.1.1. részben bemutatott stabilitásvizsgálattal összhangban válasszuk az egyik cég maximális kibocsátását a másik cég maximális kibocsátásának több, mint 7.97-szeresére. Ekkor a Cournot-megoldás instabillá válik.